3

Sistemas de ecuaciones lineales
Eliminación gaussiana
Factorización LU

Objetivos

- Saber utilizar instrucciones de *Maxima* para triangularizar y factorizar matrices y resolver sistemas de ecuaciones lineales.
- Construir una función en *Maxima* en la que se programe el algoritmo de eliminación de Gauss, con y sin estrategia de pivotaje.
- Saber calcular la factorización LU de una matriz utilizando las matrices elementales.
- Aplicar la factorización LU en la resolución de sistemas lineales.

3.1. La eliminación de Gauss como un programa en *Maxima*

El objetivo de esta sección es el de construir una función en *Maxima* con la cual automaticemos el método de resolución de sistemas mediante la eliminación de Gauss (sin permutación de filas). Antes de nada, vamos a entender qué se hace en cada iteración del proceso de eliminación gaussiana. Considerando la matriz ampliada $Ab = (A|\mathbf{b})$ asociada al sistema, el primer paso consiste en conseguir ceros en la primera columna por debajo de la posición diagonal. Esto es:

$\boxed{\text{Para } k = 1}$

Si $a_{11} \neq 0$ (pivote), para conseguir 0 en la posición $(i, 1)$, a la fila $i-$ésima le sumamos la primera fila multiplicada por $-(a_{i1}/a_{11})$. Obviamente, el resto de los coeficientes de la fila $i-$ésima se verán modificados por esta operación. De modo que,

$$\left\{\begin{array}{l} \text{para } i = 2, \ldots, m \\ \qquad mult \ = \ -a_{i1}/a_{11} \\ \qquad Ab(i)^1 \ = \ Ab(i) \ + \ mult\, Ab(1) \end{array}\right.$$

donde $Ab(i)^1$ es la fila obtenida al hacer ceros en la posición $(i, 1) de Ab$. Por tanto, tras la primera iteración tenemos

$$\begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} & b_1 \\ a_{21} & a_{22} & \dots & a_{2n} & b_2 \\ \vdots & \vdots & \ddots & \vdots & \vdots \\ a_{m1} & a_{m2} & \dots & a_{mn} & b_m \end{pmatrix} \sim \left(\begin{array}{cc} & a_{11} \quad a_{12} \quad \dots \quad a_{1n} \quad b_1 \\ \boxed{\begin{matrix} 0 \\ \vdots \\ 0 \end{matrix}} & \underbrace{\boxed{\begin{matrix} a_{22}^1 & \dots & a_{2n}^1 & b_2^1 \\ \vdots & \ddots & \vdots & \vdots \\ a_{m1}^1 & \dots & a_{mn}^1 & b_m^1 \end{matrix}}}_{Ab^1} \end{array} \right).$$

Ahora nos centraríamos en el bloque Ab^1, la parte activa de la matriz de coeficientes, de tamaño $(m-1)\times(n-1)$, y repetiríamos de forma iterada el proceso anterior, tomando ahora a_{22}^1 como pivote. Realizaríamos este razonamiento iterativamente (otras $m-2$ veces) hasta conseguir una matriz triangular superior.

A continuación se detalla el código de una nueva función de *Maxima*, `elim_gauss(A,b)`, en la que se hemos programado el **método de Gauss sin intercambio de filas**. En primer lugar, léelo atentamente y trata de comprender todos los pasos para tener claro qué se está haciendo en el programa.

`elim_gauss.mac`

```
elim_gauss(A,b):=/* Esta función resuelve el sistema Ax=b utilizando */
                 /* eliminación gaussiana, SIN intercambiar filas.   */
                 /* Argumentos de entrada:                            */
                 /*     A, matriz de coeficientes del sistema;        */
                 /*     b, vector de términos independientes.         */
    block([m,n,i,j,mult,x], /* Definimos las variables locales a usar.*/
    [m,n]:matrix_size(A), /* Calculamos el tamaño de la matriz A. */
    Ab:addcol(A,b),/* Construimos la matriz ampliada. */

    for j:1 step 1 thru n do( /* Recorremos las columnas. */
        for i:j+1 step 1 thru m do(
          /* Bajamos por debajo de la diagonal en la columna i. */
            mult:-Ab[i,j]/Ab[j,j], /* Calculamos el multiplicador. */
            Ab[i]:Ab[i]+mult*Ab[j] /* Hacemos 0 el valor de Ab[i,j]. */
            ) /* Cerramos el for que recorre las filas. */
        ), /* Cerramos el for que recorre las columnas. */
    U:submatrix(Ab,n+1), /* Extraemos la matriz de coeficientes escalonada. */
    d:col(Ab,n+1), /* Extraemos el vector de términos independientes. */
    x:triang_sup(U,d),/* Resolvemos el sistema triangular superior equivalente.*/
    x   /* Argumento de salida: x, vector solución. */
); /* Cerramos block que define la función. */
```

Observa que si alguno de los pivotes fuera nulo, el proceso se detendría al calcular el multiplicador correspondiente (`mult:-Ab[i,j]/Ab[j,j]`).

Para resolver el sistema $A\mathbf{x} = \mathbf{b}$, se carga el fichero `elim_gauss.mac`, se definen la matriz de coeficientes, A y el vector de términos independientes, $\mathbf{b}$, y se ejecuta `x:elim_gauss(A,b)`. Aplícalo en el siguiente sistema.

Ejercicio propuesto

1. Utiliza la función `elim_gauss` para resolver el siguiente sistema lineal y comprueba la

validez de la solución obtenida.

$$\begin{cases} 2x+3y-t=0 \\ x+2y-z=2 \\ y-t=-2 \\ -2x+2z+3t=6 \end{cases}$$

3.2. Factorización LU

La factorización LU de una matriz A consiste en descomponer la matriz A como producto de dos matrices

$$A = LU$$

donde L es triangular inferior con elementos diagonales $\ell_{ii}=1$ y U es una matriz triangular superior. La matriz U es la matriz triangular superior obtenida tras aplicar el proceso de eliminación de Gauss a la matriz A. De modo que, si denotamos por P_1, P_2,..., P_r a las matrices elementales utilizadas en el proceso de triangulación de la matriz A, se tiene que

$$P_r \cdots P_2 P_1 A = U,$$

luego

$$A = (P_1^{-1})(P_2^{-1})\cdots(P_r^{-1})U.$$

Si cada una de estas matrices elementales utilizadas son del tipo $P_{ij}(t)$ con $i>j$ (triangulares inferiores con 1's en la diagonal), entonces sus inversas también son triangulares inferiores con 1's en la diagonal (notar que $(P_{ij}(t))^{-1}=P_{ij}(-t)$) y, por tanto, tenemos

$$L = (P_1^{-1})(P_2^{-1})\cdots(P_r^{-1}).$$

Observación: Es importante destacar que esta factorización existe siempre que al aplicar la técnica de eliminación de Gauss no aparecezcan pivotes nulos en el proceso. De hecho, la factorización LU se puede interpretar como "la versión matricial" de la eliminación gaussiana, sin intercambio de filas, ni reescalados.

Por ejemplo:

```
(%i1) load("C:/Users/usuario/Documents/.../matrices_elementales.mac")$

(%i2) A:matrix([4,-1,0,0],[-1,3,1,0],[0,1,5,-2],[0,0,-2,3]);
```

$$(\%o2) \quad \begin{pmatrix} 4 & -1 & 0 & 0 \\ -1 & 3 & 1 & 0 \\ 0 & 1 & 5 & -2 \\ 0 & 0 & -2 & 3 \end{pmatrix}$$

```
(%i3) A1:pijt(2,1,1/4,4).A;
```

$$(\%o3) \quad \begin{pmatrix} 4 & -1 & 0 & 0 \\ 0 & \frac{11}{4} & 1 & 0 \\ 0 & 1 & 5 & -2 \\ 0 & 0 & -2 & 3 \end{pmatrix}$$

```
(%i4) A2:pijt(3,2,-4/11,4).A1;
```

$$(\%o4) \quad \begin{pmatrix} 4 & -1 & 0 & 0 \\ 0 & \frac{11}{4} & 1 & 0 \\ 0 & 0 & \frac{51}{11} & -2 \\ 0 & 0 & -2 & 3 \end{pmatrix}$$

```
(%i5) U:pijt(4,3,22/51,4).A2;
```

$$(\%o5)\quad \begin{pmatrix} 4 & -1 & 0 & 0 \\ 0 & \frac{11}{4} & 1 & 0 \\ 0 & 0 & \frac{51}{11} & -2 \\ 0 & 0 & 0 & \frac{109}{51} \end{pmatrix}$$

```
(%i6) L:pijt(2,1,-1/4,4).pijt(3,2,4/11,4).pijt(4,3,-22/51,4);
```

$$(\%o6)\quad \begin{pmatrix} 1 & 0 & 0 & 0 \\ -\frac{1}{4} & 1 & 0 & 0 \\ 0 & \frac{4}{11} & 1 & 0 \\ 0 & 0 & -\frac{22}{51} & 1 \end{pmatrix}$$

```
(%i7) is(A=L.U);
```

(%o7) *true*

Ejercicios propuestos

2. Procede de forma análoga al ejemplo anterior para calcular la factorización LU de la matriz
$$A = \begin{pmatrix} 1 & -2 & 1 \\ 1 & 1 & 3 \\ -1 & 2 & 1 \end{pmatrix}.$$
Comprueba que la factorización obtenida es correcta: L y U tienen la forma adecuada y su producto es la matriz A.

3. Si una matriz cuadrada A admite factorización LU, ¿qué relación existe entre el determinante de A y los determinantes de L y U? Teniendo en cuenta la respuesta que acabas de dar, calcula el determinante de la matriz del ejercicio anterior. (Comprueba que coincide con el valor que devuelve la orden `determinant(A)`).

Recuerda que la factorización LU se puede aplicar para resolver sistemas de ecuaciones lineales $A\mathbf{x} = \mathbf{b}$ de la siguiente forma:

$$A\,\mathbf{x} = \mathbf{b} \;\Leftrightarrow\; L\underbrace{U\,\mathbf{x}}_{\mathbf{y}} = \mathbf{b} \;\Leftrightarrow\; \left\{ \begin{array}{lcll} L\,\mathbf{y} & = & \mathbf{b} & \text{(triangular inf.)} \\ U\,\mathbf{x} & = & \mathbf{y} & \text{(triangular sup.)} \end{array} \right.$$

Ejercicios propuestos

4. Utiliza las funciones `triang_sup` y `triang_inf` implementadas en la práctica anterior para volver a resolver el sistema del ejercicio 1. utilizando la factorización LU de su matriz de coeficientes.

5. Utiliza la función `elim_gauss` para resolver con el método de Gauss el siguiente sistema:
$$\begin{pmatrix} 2 & 1 & 2 & 0 \\ -4 & -2 & 1 & 3 \\ 1 & 1/2 & 2 & 3 \\ 4 & 0 & 9 & 3 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{pmatrix} = \begin{pmatrix} 6 \\ -4 \\ 7 \\ 20 \end{pmatrix}$$
¿Qué ocurre? ¿Por qué?

3.3. Estrategia de pivotaje parcial

Es sabido que el método básico de eliminación gaussiana falla si alguno de los pivotes del proceso es nulo. En este apartado planteamos una versión de la eliminación gaussiana que incorpora la *estrategia de pivotaje parcial* con la cual se evitan los pivotes nulos. Además, se corrige otro de los inconvenientes de este método, como es la inestabilidad del algoritmo debido a los errores de redondeo.

Esta variante consiste, en lugar de tomar, por defecto, como pivote el elemento diagonal a_{kk}^{k-1}, elegir como pivote el coeficiente de mayor valor absoluto en la columna desde la diagonal hacia abajo, reordenando convenientemente las filas. Veámoslo sobre un ejemplo concreto. Considera la matriz

$$Ab = \begin{pmatrix} 1 & 2 & 1 & 4 & 8 \\ 0 & -2 & 3 & 4 & 5 \\ -5 & 3 & 1 & 4 & 3 \\ 4 & 1 & 2 & 3 & 10 \end{pmatrix}.$$

Al aplicar la primera vez la estrategia de pivotaje parcial sobre esta matriz, obtendríamos como primer pivote: $Ab_{31} = -5$, pues es elemento bajo la posición (1,1) de mayor valor absoluto. Una vez localizado este pivote y la fila en la que se encuentra, se reordenan las filas de la matriz para conseguir una matriz equivalente sobre la que aplicar la eliminación gaussiana. A continuación se presenta como se haría con *Maxima:*

```
(%i8) Ab:matrix([1,2,1,4,8],[0,-2,3,4,5],[-5,3,1,4,3],[4,1,2,3,10]);
```

(%o8) $\begin{pmatrix} 1 & 2 & 1 & 4 & 8 \\ 0 & -2 & 3 & 4 & 5 \\ -5 & 3 & 1 & 4 & 3 \\ 4 & 1 & 2 & 3 & 10 \end{pmatrix}$

```
(%i9) [n,m]:matrix_size(Ab);
```

(%o9) $[4, 5]$

```
(%i10) pivote:Ab[1,1];
```

(%o10) 1

```
(%i11) fila:1;
```

(%o11) 1

```
(%i12) for i:2 step 1 thru n do(
           if (abs(Ab[i,1]) > pivote) then (
               pivote:abs(Ab[i,1]),
               fila:i
           )
       )$
```

```
(%i13) pivote;
```

(%o13) 5

```
(%i14) fila;
```

(%o14) 3

```
(%i15) if (fila > 1) then (
              Ab:rowswap(Ab,fila,1)
          );
```

$$(\%o15)\quad \begin{pmatrix} -5 & 3 & 1 & 4 & 3 \\ 0 & -2 & 3 & 4 & 5 \\ 1 & 2 & 1 & 4 & 8 \\ 4 & 1 & 2 & 3 & 10 \end{pmatrix}$$

Esta estrategia de pivotaje parcial está parcialmente implementada en el fichero `gauss_pivote.mac`. Abre el fichero y léelo.

Ejercicios propuestos

6. Completa la función `gauss_pivote(A,b)` y comprueba que funciona correctamente resolviendo el sistema del ejercicio anterior.

7. Resuelve los siguientes sistemas aplicando la eliminación gaussiana utilizando la función programada `elim_gauss(A,b)` o `gauss_pivote(A,b)`, según convenga.

   a) $\begin{pmatrix} 2 & 1 & 1 & 1 \\ 3 & 1 & 0 & 0 \\ 7 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{pmatrix} = \begin{pmatrix} -1 \\ 2 \\ 0 \\ 0 \end{pmatrix}$, b) $\begin{pmatrix} 0 & 2 & 1 & -2 \\ 3 & 0 & 0 & 1 \\ 7 & 0 & 2 & 0 \\ 0 & 1 & 0 & 2 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{pmatrix} = \begin{pmatrix} 2 \\ 1 \\ 1 \\ 2 \end{pmatrix}$.

   c) $\begin{pmatrix} 4 & 2 & 0 & 0 \\ 2 & 1 & 1 & 1 \\ 7 & 0 & -1 & 0 \\ -3 & 1 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{pmatrix} = \begin{pmatrix} 6 \\ 5 \\ 6 \\ -1 \end{pmatrix}$, d) $\begin{pmatrix} 5 & 1 & 0 & 3 \\ 3 & 0 & 1 & 1 \\ 1 & 0 & 1 & 0 \\ 2 & 1 & 1 & 0 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{pmatrix} = \begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}$.

8. Halla la factorización LU de la matriz de coeficientes de los sistemas del ejercicio anterior, reordenando previamente las filas si fuera necesario, y expresando la matriz L como producto de matrices elementales.

9. Utiliza las factorizaciones halladas en el ejercicio anterior para resolver los sistemas lineales correspondientes.
