1 Matrices y aplicaciones del cálculo matricial

Objetivos

- Utilizar las instrucciones básicas de *Maxima* para definir, visualizar, operar y manipular matrices con soltura.
- Utilizar las funciones de *Maxima* para hallar inversas de matrices, calcular el rango de una matriz, su traspuesta, determinantes, ...
- Construir matrices elementales y comprender la relación que existe con las operaciones elementales sobre matrices.
- Conocer distintas aplicaciones de las matrices.

1.1. Matrices

1.1.1. Definición de matrices

Para introducir matrices en *Maxima* utilizaremos la función matrix. Esta función tiene como argumentos las filas de la matriz escritas como listas. Por ejemplo:

```
(%i1) A:matrix([a,b,c],[d,e,f]);
```

(%o1) $\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}$

También es posible introducir matrices en *Maxima*, usando el menú Álgebra → Introducir matriz. El primer cuadro de diálogo que aparece da la posibilidad de introducir el tamaño de la matriz, así como el tipo y el nombre elegidos para la matriz. Una vez especificados estos datos, se abre un segundo cuadro en el que se deben especificar los elementos de la matriz (ver Figura 1.1).



Figura 1.1: Cuadros de diálogo al introducir una matriz desde el menú principal

1.1.2. Operaciones con matrices

Una vez que ya sabemos cómo definir matrices, veamos qué comandos nos permiten operar con ellas.

$+$	suma de matrices del mismo tamaño
$-$	resta de matrices del mismo tamaño
.	producto matricial de matrices de tamaños compatibles
$*$	producto de un escalar por una matriz
^^ n	potencia $n-$ésima de una matriz cuadrada

Por ejemplo:

```
(%i2) B:matrix([1,1,1],[-1,0,3]);
```

(%o2) $\begin{pmatrix} 1 & 1 & 1 \\ -1 & 0 & 3 \end{pmatrix}$

```
(%i3) C:matrix([2,-8,7],[-3,9,6]);
```

(%o3) $\begin{pmatrix} 2 & -8 & 7 \\ -3 & 9 & 6 \end{pmatrix}$

```
(%i4) B+C;
```

(%o4) $\begin{pmatrix} 3 & -7 & 8 \\ -4 & 9 & 9 \end{pmatrix}$

```
(%i5) B.C;
```

MULTIPLYMATRICES: attempt to multiply nonconformable matrices. – an error. To debug this try: debugmode(true);

Recuerda que para que esta operación sea posible el número de columnas de la primera matriz ha de coincidir con el número de filas de la segunda matriz. Por ello, en este caso observamos que *Maxima* devuelve un mensaje indicando que no tiene sentido realizar ese producto matricial.

En el siguiente ejemplo los tamaños son compatibles y el producto sí es posible:

```
(%i6) D:matrix([-1,2,-3],[0,1,4],[1,-2,9]);
```

(%o6) $\begin{pmatrix} -1 & 2 & -3 \\ 0 & 1 & 4 \\ 1 & -2 & 9 \end{pmatrix}$

```
(%i7) B.D;
```

(%o7) $\begin{pmatrix} 0 & 1 & 10 \\ 4 & -8 & 30 \end{pmatrix}$

El operador * se utiliza para multiplicar un escalar por una matriz. Por ejemplo:

```
(%i8) 2*B;
```

(%o8) $\begin{pmatrix} 2 & 2 & 2 \\ -2 & 0 & 6 \end{pmatrix}$

Observación. Es importante remarcar que el operador * también puede usarse para realizar el *producto elemento a elemento* de dos matrices del mismo tamaño:

```
(%i9)   B*C;
```

(%o9) $\begin{pmatrix} 2 & -8 & 7 \\ 3 & 0 & 18 \end{pmatrix}$

Para calcular la potencia $n-$ésima de una matriz utilizaremos el operador $\hat{}\,\hat{}\,n$. Por ejemplo, para cacular D^2 escribiremos:

```
(%i10)  D^^2;
```

(%o10) $\begin{pmatrix} -2 & 6 & -16 \\ 4 & -7 & 40 \\ 8 & -18 & 70 \end{pmatrix}$

Obviamente el resultado es el mismo que se obtiene al ejecutar el producto matricial de la matriz D por ella misma:

```
(%i11) D.D;
```

(%o11) $\begin{pmatrix} -2 & 6 & -16 \\ 4 & -7 & 40 \\ 8 & -18 & 70 \end{pmatrix}$

Observación. No hay que confundir el comando anterior con el símbolo $\hat{}\,n$ que eleva a la potencia n cada elemento de la matriz:

```
(%i12)  D^2;
```

(%o12) $\begin{pmatrix} 1 & 4 & 9 \\ 0 & 1 & 16 \\ 1 & 4 & 81 \end{pmatrix}$

Ejercicio propuesto

1. Definir en *Maxima* las matrices:

$$M_1 = \begin{pmatrix} 1 & 0 \\ 3 & -2 \end{pmatrix}, \quad M_2 = \begin{pmatrix} 2 & 3 \\ 5 & 0 \\ 1 & 2 \end{pmatrix}, \quad M_3 = \begin{pmatrix} 2 \\ -1 \end{pmatrix}, \quad M_4 = \begin{pmatrix} 2 & -1 \end{pmatrix}, \quad M_5 = \begin{pmatrix} 0 & -2 \\ -4 & -1 \end{pmatrix}.$$

   a) Calcular las siguientes operaciones, siempre que sea posible:

$$2M_1 + 3M_5;, \qquad M_3 - 2M_4; \qquad {M_2}^2 \quad \text{(cuadrado de la matriz } M_2\text{)}.$$

   b) Realizar todos los productos matriciales posibles tomando matrices de 2 en 2.

   c) Comprobar que el producto de matrices no es conmutativo.

   d) Explicar los resultados que se obtienen al ejecutar los productos $\mathrm{M}_1.\mathrm{M}_1$ y $\mathrm{M}_1 * \mathrm{M}_1$.

   e) Hallar una matriz 3×2 donde el elemento (i, j) es el cubo del elemento (i, j) de la matriz M_2.

1.1.3. Otras operaciones con matrices

Maxima permite realizar otras operaciones con matrices.

invert(A)	calcula la inversa de A
transpose(A)	calcula la traspuesta de A
rank(A)	calcula el rango de A
determinant(A)	calcula el determinante de A
triangularize(A)	calcula una matriz triangular superior equivalente a A
matrix_size(A)	devuelve el tamaño de A

Por ejemplo:

```
(%i13) D;
```

(%o13) $$\begin{pmatrix} -1 & 2 & -3 \\ 0 & 1 & 4 \\ 1 & -2 & 9 \end{pmatrix}$$

```
(%i14) invD:invert(D);
```

(%o14) $$\begin{pmatrix} -\frac{17}{6} & 2 & -\frac{11}{6} \\ -\frac{2}{3} & 1 & -\frac{2}{3} \\ \frac{1}{6} & 0 & \frac{1}{6} \end{pmatrix}$$

Podemos comprobarlo verificando que $DD^{-1} = I$:

```
(%i15) D.invD;
```

(%o15) $$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i16) transpose(D);
```

(%o16) $$\begin{pmatrix} -1 & 0 & 1 \\ 2 & 1 & -2 \\ -3 & 4 & 9 \end{pmatrix}$$

```
(%i17) rank(D);
```

(%o17) 3

El rango de la matriz D es 3. También podemos comprobarlo con la orden triangularize.

```
(%i18) triangularize(D);
```

(%o18) $$\begin{pmatrix} -1 & 2 & -3 \\ 0 & -1 & -4 \\ 0 & 0 & -6 \end{pmatrix}$$

La orden matrix_size devuelve el tamaño de la matriz:

```
(%i19) matrix_size(D);
```

(%o19) $[3, 3]$

Ejercicios propuestos

2. Considera la matriz:

$$A = \begin{pmatrix} a+1 & 1 & 1 \\ 1 & a-1 & 1 \\ 0 & 1 & a+2 \end{pmatrix}$$

   a) A partir de $|A|$, encontrar los valores de a para los que A es regular.
   b) ¿Qué ocurre si usas la orden `rank(A)`?
   c) Para el caso particular $a = 0$, hallar la inversa de A.

3. Dada la matriz:

$$A = \begin{pmatrix} 1 & 1 & 1 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix},$$

   calcular las potencias A^2, A^3, A^4 y A^5. ¿Puedes deducir cuál es la potencia A^n?

1.1.4. Introducción de matrices particulares

Maxima permite gererar algunos tipos particulares de matrices de forma sencilla.

ident(n)	matriz identidad de orden n
zeromatrix(n,m)	matriz de tamaño $n \times m$ con todas las entradas nulas
diagmatrix(n,x)	matriz diagonal $n \times n$ con x en la diagonal

(%i20) `ident(3);`

(%o20) $\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

(%i21) `zeromatrix(2,3);`

(%o21) $\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$

(%i22) `diagmatrix(2,-3);`

(%o22) $\begin{pmatrix} -3 & 0 \\ 0 & -3 \end{pmatrix}$

Otra forma de construir matrices diagonales, con elementos diagonales no necesariamente iguales, es usando el menú Álgebra → Introducir matriz y en el cuadro de diálogo elegir como el tipo de matriz **diagonal** y especificar los elementos de la matriz (ver Figura 1.2).



Figura 1.2: Cuadros de diálogo al introducir una matriz diagonal desde el menú

Ejercicio propuesto

4. Considerar la matriz:

$$A = \begin{pmatrix} 1 & 1 & 1 \\ 0 & 2 & 0 \\ 0 & 1 & 2 \end{pmatrix}.$$

   a) Calcular el polinomio $|xI_3 - A|$.
   b) Comprobar que $x_1 = 1$ y $x_2 = 2$ son raíces del polinomio anterior.
   c) Triangularizar las matrices $M1 = I_3 - A$ y $M2 = 2I_3 - A$ y hallar sus respectivos rangos.

1.1.5. Manipulación de matrices

Maxima permite la manipulación de matrices a través de distintos comandos.

A[i,j]	elemento ij de A
col(A,n)	columna n de A
row(A,n)	fila n de A
addrow(A,lista1,lista2,...)	añade a la matriz A las filas lista1, lista2, ...
addcol(A,lista1,lista2,...)	añade a la matriz A las columnas lista1, lista2, ...
submatrix($f_1, f_2, \ldots$,A,$c_1, c_2, \ldots$)	submatriz obtenida a partir de A al eliminar las filas $f_1, f_2, \ldots$ y las columnas $c_1, c_2, \ldots$
load(diag)	carga el paquete diag
diag([A1,A2,...])	construye una matriz diagonal por bloques

Por ejemplo:

```
(%i23) M1:matrix([1,2,0],[6,0,-2],[0,-5,a]);
```

(%o23) $$\begin{pmatrix} 1 & 2 & 0 \\ 6 & 0 & -2 \\ 0 & -5 & a \end{pmatrix}$$

Si nos interesa extraer el elemento $(3,2)$ de la matriz $M1$ usaremos la orden:

```
(%i24) M1[3,2];
```

(%o24) -5

Para extraer de $M1$ la submatriz que resulta de eliminar la primera fila y la tercera columna, escribiremos:

```
(%i25) submatrix(1,M1,3);
```

(%o25) $$\begin{pmatrix} 6 & 0 \\ 0 & -5 \end{pmatrix}$$

Si queremos eliminar la tercera fila y las dos primeras columnas:

```
(%i26) submatrix(3,M1,1,2);
```

(%o26) $$\begin{pmatrix} 0 \\ -2 \end{pmatrix}$$

Podemos ampliar el número de filas y de columnas de una matriz dada:

```
(%i27)  addrow(M1,[0,-1,2],[9,1,-3]);
```

(%o27) $\begin{pmatrix} 1 & 2 & 0 \\ 6 & 0 & -2 \\ 0 & -5 & a \\ 0 & -1 & 2 \\ 9 & 1 & -3 \end{pmatrix}$

```
(%i28)  addcol(%,[-2,0,0,7,9]);
```

(%o28) $\begin{pmatrix} 1 & 2 & 0 & -2 \\ 6 & 0 & -2 & 0 \\ 0 & -5 & a & 0 \\ 0 & -1 & 2 & 7 \\ 9 & 1 & -3 & 9 \end{pmatrix}$

```
(%i29) addrow(M1,ident(3));
```

(%o29) $\begin{pmatrix} 1 & 2 & 0 \\ 6 & 0 & -2 \\ 0 & -5 & a \\ 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

```
(%i30) addrow(ident(3),zeromatrix(3,3));
```

(%o30) $\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$

Podemos evaluar la matriz $M1$ para un valor concreto del parámetro a:

```
(%i31)  M1,a=1;
```

(%o31) $\begin{pmatrix} 1 & 2 & 0 \\ 6 & 0 & -2 \\ 0 & -5 & 1 \end{pmatrix}$

Para construir una matriz diagonal por bloques se precisa cargar previamente el paquete `diag`:

```
(%i32) load(diag)$
(%i33) A2:matrix([2,2],[2,2])$
(%i34) A3:matrix([3,3,3],[3,3,3],[3,3,3])$
(%i35) A:diag([A2,A3]);
```

(%o35) $\begin{pmatrix} 2 & 2 & 0 & 0 & 0 \\ 2 & 2 & 0 & 0 & 0 \\ 0 & 0 & 3 & 3 & 3 \\ 0 & 0 & 3 & 3 & 3 \\ 0 & 0 & 3 & 3 & 3 \end{pmatrix}$

Ejercicio propuesto

5. Definir las matrices

$$A_1 = \begin{pmatrix} 1 & 1 & 1 \\ 1 & -1 & 1 \\ 0 & 1 & 2 \end{pmatrix}, \qquad A_2 = \begin{pmatrix} 1 & 1 \\ 0 & 4 \end{pmatrix}.$$

   a) A partir de A_1 y A_2 definir la matriz diagonal por bloques

   $$A = \begin{pmatrix} A_1 & O \\ O & A_2 \end{pmatrix}.$$

   b) Deducir la relación entre el rango de A y los rangos de las matrices A_1 y A_2.

   c) Relacionar el valor del determinante de A con el valor de los determinantes de las matrices A_1 y A_2.

   d) A partir de A, recuperar las matrices A_1 y A_2 eliminando, convenientemente, filas y columnas de A.

1.1.6. Matrices elementales

Si bien *Maxima* dispone de instrucciones que recogen dos de las tres operaciones elementales por filas:

rowswap(M,i,j)	intercambia las filas i y j de la matriz M
rowop(M,i,j,a)	sustituye la fila i por la fila que resulta de realizar la operación $F_i \to F_i - aF_j$

las siguientes funciones permiten construir los tres tipos de matrices elementales y utilizarlas para realizar operaciones elementales, tanto por filas como por columnas.

Matriz permutación P_{ij}:

```
(%i36) pij(i,j,n):=block([P],
       /* i, j marcan posiciones de elementos que difieren de la identidad */
       /* n representa el tamaño de la matriz */
       P:ident(n),
       P[i,i]:0,
       P[i,j]:1,
       P[j,i]:1,
       P[j,j]:0,
       return(P)
       )$
```

```
(%i37) A:matrix([1,1,1,1],[2,2,2,0],[3,3,0,0],[4,0,0,0]);
```

$$(\%o37) \quad \begin{pmatrix} 1 & 1 & 1 & 1 \\ 2 & 2 & 2 & 0 \\ 3 & 3 & 0 & 0 \\ 4 & 0 & 0 & 0 \end{pmatrix}$$

```
(%i38) pij(1,2,4).A  /*  permutamos las filas 1 y 2 de A */;
```

$$(\%o38)\quad \begin{pmatrix} 2 & 2 & 2 & 0 \\ 1 & 1 & 1 & 1 \\ 3 & 3 & 0 & 0 \\ 4 & 0 & 0 & 0 \end{pmatrix}$$

```
(%i39) A.pij(1,3,4)  /* permutamos las columnas 1 y 3 de A */;
```

$$(\%o39)\quad \begin{pmatrix} 1 & 1 & 1 & 1 \\ 2 & 2 & 2 & 0 \\ 0 & 3 & 3 & 0 \\ 0 & 0 & 4 & 0 \end{pmatrix}$$

Matriz $P_{ij}(t),\ t \in \mathbb{R}$**:**

```
(%i40) pijt(i,j,t,n):=block([P],
       /* (i, j) posición en la que se encuentra el escalar t */
       /* n representa el tamaño de la matriz P */
       P:ident(n),
       P[i,j]:t,
       return(P)
       )$
```

```
(%i41) pijt(2,3,-5,4);
```

$$(\%o41)\quad \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & -5 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(%i42) pijt(2,1,-2,4).A  /* F2 + (-2)*F1 */;
```

$$(\%o42)\quad \begin{pmatrix} 1 & 1 & 1 & 1 \\ 0 & 0 & 0 & -2 \\ 3 & 3 & 0 & 0 \\ 4 & 0 & 0 & 0 \end{pmatrix}$$

```
(%i43) A.pijt(1,2,-1,4)  /* C2 + (-1)*C1 */;
```

$$(\%o43)\quad \begin{pmatrix} 1 & 0 & 1 & 1 \\ 2 & 0 & 2 & 0 \\ 3 & 0 & 0 & 0 \\ 4 & -4 & 0 & 0 \end{pmatrix}$$

Matriz $Q_i(s),\ s \neq 0$**.**

```
(%i44) qis(i,s,n):=block([P],
       /* (i,i) elemento diagonal igual al escalar s */
       /* n representa el tamaño de la matriz P */
       P:ident(n),
       P[i,i]:s,
       return(P)
       )$
```

```
(%i45) qis(1,-3,4);
```

$$(\%o45)\quad \begin{pmatrix} -3 & 0 & 0 & 0\\ 0 & 1 & 0 & 0\\ 0 & 0 & 1 & 0\\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(%i46) qis(2,2,4).A  /* Multiplicamos la fila 2 por 2 */;
```

$$(\%o46)\quad \begin{pmatrix} 1 & 1 & 1 & 1\\ 4 & 4 & 4 & 0\\ 3 & 3 & 0 & 0\\ 4 & 0 & 0 & 0 \end{pmatrix}$$

```
(%i47) A.qis(3,-1,4)  /* Multiplicamos la columna 3 por (-1) */;
```

$$(\%o47)\quad \begin{pmatrix} 1 & 1 & -1 & 1\\ 2 & 2 & -2 & 0\\ 3 & 3 & 0 & 0\\ 4 & 0 & 0 & 0 \end{pmatrix}$$

Observación. Es importante remarcar que para utilizar estas nuevas funciones definidas en *Maxima* no es necesario teclear y ejecutar todas las órdenes cada vez. Éstas las hemos guardado en el fichero `matrices_elementales.mac` y es suficiente cargar este paquete para poder usarla como cualquiera de las funciones que *Maxima* tiene definidas.

En el menú pricipal seleccionar: `Archivo ->Cargar paquete` y buscar el fichero en el ordenador.

Ejercicio propuesto

6. Sobre la matriz

$$A = \begin{pmatrix} 2 & 1 & 1\\ 1 & -1 & 3\\ -2 & 1 & 0 \end{pmatrix}$$

realizar las operaciones elementales que se indican a continuación (utilizando las funciones anteriormente definidas):

   a) A la fila 2 sumarle la fila 3 multiplicada por -3.
   b) A la columna 1 sumarle la columna 2 multiplicada por 4.
   c) Permutar las columnas 1 y 2.
   d) Multiplicar la fila 2 por -1.

1.1.7. Matrices equivalentes. Rango de una matriz

Una vez definidas las funciones en *Maxima* que nos permiten construir las matrices elementales, veamos cómo utilizarlas para transformar una matriz en otra matriz escalonada equivalente y así calcular el rango.

Podemos calcular el rango de una matriz haciendo operaciones elementales sobre ésta:

```
(%i48) A: matrix([2,1,-3,4],[4,0,-1,2],[4,2,-1,2]);
```

(%o48) $\begin{pmatrix} 2 & 1 & -3 & 4 \\ 4 & 0 & -1 & 2 \\ 4 & 2 & -1 & 2 \end{pmatrix}$

```
(%i49) A1:pijt(2,1,-2,3).A;
```

(%o49) $\begin{pmatrix} 2 & 1 & -3 & 4 \\ 0 & -2 & 5 & -6 \\ 4 & 2 & -1 & 2 \end{pmatrix}$

```
(%i50) A2:pijt(3,1,-2,3).A1;
```

(%o50) $\begin{pmatrix} 2 & 1 & -3 & 4 \\ 0 & -2 & 5 & -6 \\ 0 & 0 & 5 & -6 \end{pmatrix}$

Luego $\operatorname{rang}(A) = 3$ como podemos comprobar con *Maxima*:

```
(%i51) rank(A);
```

(%o51) 3

También podemos encontrar matrices regulares $P \in \mathcal{M}_3(\mathbb{R})$ y $Q \in \mathcal{M}_4(\mathbb{R})$ que establecen la equivalencia entre la matriz A y la matriz

$$C = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \end{pmatrix},$$

con $C = PAQ$. La matriz P se construye a partir de las matrices elementales utilizadas en las operaciones por filas y la matriz Q resulta del producto matricial de las matrices elementales usadas en las operaciones por columnas. Es obvio, que estas matrices P y Q no son únicas, pues no existe una secuencia única de operaciones elementales (por filas y/o columnas) para transformar A en la matriz equivalente C. Veamos una posible forma de hacerlo:

```
(%i52) A1:qis(1,1/2,3).A;
```

(%o52) $\begin{pmatrix} 1 & \frac{1}{2} & -\frac{3}{2} & 2 \\ 4 & 0 & -1 & 2 \\ 4 & 2 & -1 & 2 \end{pmatrix}$

```
(%i53) P1:qis(1,1/2,3)$
```

```
(%i54) A2:pijt(2,1,-4,3).A1;
```

(%o54) $\begin{pmatrix} 1 & \frac{1}{2} & -\frac{3}{2} & 2 \\ 0 & -2 & 5 & -6 \\ 4 & 2 & -1 & 2 \end{pmatrix}$

```
(%i55) P2:pijt(2,1,-4,3)$
```

```
(%i56) A3:pijt(3,1,-4,3).A2;
```

(%o56) $\begin{pmatrix} 1 & \frac{1}{2} & -\frac{3}{2} & 2 \\ 0 & -2 & 5 & -6 \\ 0 & 0 & 5 & -6 \end{pmatrix}$

```
(%i57) P3:pijt(3,1,-4,3)$
```

(%i58) `A4:qis(2,-1/2,3).A3;`

$$(\%o58)\quad \begin{pmatrix}1 & \frac{1}{2} & -\frac{3}{2} & 2\\ 0 & 1 & -\frac{5}{2} & 3\\ 0 & 0 & 5 & -6\end{pmatrix}$$

(%i59) `P4:qis(2,-1/2,3)$`

(%i60) `A5:qis(3,1/5,3).A4;`

$$(\%o60)\quad \begin{pmatrix}1 & \frac{1}{2} & -\frac{3}{2} & 2\\ 0 & 1 & -\frac{5}{2} & 3\\ 0 & 0 & 1 & -\frac{6}{5}\end{pmatrix}$$

(%i61) `P5:qis(3,1/5,3)$`

(%i62) `A6:A5.pijt(3,4,6/5,4);`

$$(\%o62)\quad \begin{pmatrix}1 & \frac{1}{2} & -\frac{3}{2} & \frac{1}{5}\\ 0 & 1 & -\frac{5}{2} & 0\\ 0 & 0 & 1 & 0\end{pmatrix}$$

(%i63) `Q1:pijt(3,4,6/5,4)$`

(%i64) `A7:A6.pijt(2,3,5/2,4);`

$$(\%o64)\quad \begin{pmatrix}1 & \frac{1}{2} & -\frac{1}{4} & \frac{1}{5}\\ 0 & 1 & 0 & 0\\ 0 & 0 & 1 & 0\end{pmatrix}$$

(%i65) `Q2:pijt(2,3,5/2,4)$`

(%i66) `A8:A7.pijt(1,4,-1/5,4);`

$$(\%o66)\quad \begin{pmatrix}1 & \frac{1}{2} & -\frac{1}{4} & 0\\ 0 & 1 & 0 & 0\\ 0 & 0 & 1 & 0\end{pmatrix}$$

(%i67) `Q3:pijt(1,4,-1/5,4)$`

(%i68) `A9:A8.pijt(1,3,1/4,4);`

$$(\%o68)\quad \begin{pmatrix}1 & \frac{1}{2} & 0 & 0\\ 0 & 1 & 0 & 0\\ 0 & 0 & 1 & 0\end{pmatrix}$$

(%i69) `Q4:pijt(1,3,1/4,4)$`

(%i70) `A10:A9.pijt(1,2,-1/2,4);`

$$(\%o70)\quad \begin{pmatrix}1 & 0 & 0 & 0\\ 0 & 1 & 0 & 0\\ 0 & 0 & 1 & 0\end{pmatrix}$$

(%i71) `Q5:pijt(1,2,-1/2,4)$`

(%i72) `C:A10$`

(%i73) `P:P5.P4.P3.P2.P1;`

(%o73) $\begin{pmatrix} \frac{1}{2} & 0 & 0 \\ 1 & -\frac{1}{2} & 0 \\ -\frac{2}{5} & 0 & \frac{1}{5} \end{pmatrix}$

```
(%i74) Q:Q1.Q2.Q3.Q4.Q5;
```

(%o74) $\begin{pmatrix} 1 & -\frac{1}{2} & \frac{1}{4} & -\frac{1}{5} \\ 0 & 1 & \frac{5}{2} & 0 \\ 0 & 0 & 1 & \frac{6}{5} \\ 0 & 0 & 0 & 1 \end{pmatrix}$

```
(%i75) is(P.A.Q=C);
```

(%o75) *true*

Ejercicio propuesto

7. Dadas la matrices:

$$A = \begin{pmatrix} 1 & 4 & 5 \\ 2 & 5 & 7 \\ 3 & 7 & 9 \end{pmatrix}, \quad B = \begin{pmatrix} 1 & 1 & -1 & 3 \\ -1 & 4 & 5 & -2 \\ 1 & 6 & 3 & 4 \end{pmatrix}.$$

a) Realizar operaciones elementales para transformarlas en una matriz escalonada.

b) Hallar el rango.

1.1.8. Cálculo de la inversa de una matriz regular

Podemos calcular la inversa de una matriz regular realizando operaciones elementales, por filas o por columnas (sin mezclarlas):

```
(%i76) A: matrix([2,-2,0],[1,3,1],[-3,1,1]);
```

(%o76) $\begin{pmatrix} 2 & -2 & 0 \\ 1 & 3 & 1 \\ -3 & 1 & 1 \end{pmatrix}$

```
(%i77) A1:addcol(A,ident(3));
```

(%o77) $\begin{pmatrix} 2 & -2 & 0 & 1 & 0 & 0 \\ 1 & 3 & 1 & 0 & 1 & 0 \\ -3 & 1 & 1 & 0 & 0 & 1 \end{pmatrix}$

```
(%i78) A2:qis(1,1/2,3).A1;
```

(%o78) $\begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 1 & 3 & 1 & 0 & 1 & 0 \\ -3 & 1 & 1 & 0 & 0 & 1 \end{pmatrix}$

```
(%i79) A3:pijt(2,1,-1,3).A2;
```

(%o79) $\begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 4 & 1 & -\frac{1}{2} & 1 & 0 \\ -3 & 1 & 1 & 0 & 0 & 1 \end{pmatrix}$

(%i80) `A4:pijt(3,1,3,3).A3;`

$$(\%o80)\quad \begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 4 & 1 & -\frac{1}{2} & 1 & 0 \\ 0 & -2 & 1 & \frac{3}{2} & 0 & 1 \end{pmatrix}$$

(%i81) `A5:qis(2,1/4,3).A4;`

$$(\%o81)\quad \begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 1 & \frac{1}{4} & -\frac{1}{8} & \frac{1}{4} & 0 \\ 0 & -2 & 1 & \frac{3}{2} & 0 & 1 \end{pmatrix}$$

(%i82) `A6:pijt(3,2,2,3).A5;`

$$(\%o82)\quad \begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 1 & \frac{1}{4} & -\frac{1}{8} & \frac{1}{4} & 0 \\ 0 & 0 & \frac{3}{2} & \frac{5}{4} & \frac{1}{2} & 1 \end{pmatrix}$$

(%i83) `A7:qis(3,2/3,3).A6;`

$$(\%o83)\quad \begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 1 & \frac{1}{4} & -\frac{1}{8} & \frac{1}{4} & 0 \\ 0 & 0 & 1 & \frac{5}{6} & \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

(%i84) `A8:pijt(2,3,-1/4,3).A7;`

$$(\%o84)\quad \begin{pmatrix} 1 & -1 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 1 & 0 & -\frac{1}{3} & \frac{1}{6} & -\frac{1}{6} \\ 0 & 0 & 1 & \frac{5}{6} & \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

(%i85) `A9:pijt(1,2,1,3).A8;`

$$(\%o85)\quad \begin{pmatrix} 1 & 0 & 0 & \frac{1}{6} & \frac{1}{6} & -\frac{1}{6} \\ 0 & 1 & 0 & -\frac{1}{3} & \frac{1}{6} & -\frac{1}{6} \\ 0 & 0 & 1 & \frac{5}{6} & \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

(%i86) `invA:submatrix(A9,1,2,3);`

$$(\%o86)\quad \begin{pmatrix} \frac{1}{6} & \frac{1}{6} & -\frac{1}{6} \\ -\frac{1}{3} & \frac{1}{6} & -\frac{1}{6} \\ \frac{5}{6} & \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

(%i87) `is(A.invA=ident(3));`

(%o87) *true*

Observación. En el bloque derecho inicialmente se parte de la matriz identidad, $BD_0 = I$, y se van "almacenando" las operaciones elementales por filas realizadas. Se puede comprobar que en la iteración i-ésima de este proceso, en el bloque derecho BD_i se tiene la matriz

$$BD_i \;=\; P_i\, P_{i-1}\, \ldots P_2\, P_1,$$

siendo P_i la matriz elemental asociada a la operación elemental realizada en la i-ésima iteración.

De manera que si para calcular la inversa de A se realizan un total de r operaciones elementales por filas, se tiene que:

$$A^{-1} \;=\; BD_r \;=\; P_r\, P_{r-1}\, \ldots P_2\, P_1.$$

Se puede comprobar que en el ejemplo anterior:

$$A^{-1} \;=\; P_{12}(1)\, P_{23}(-1/4)\, Q_3(2/3)\;\, P_{32}(2)\, Q_2(1/4)\, P_{31}(3)\, P_{21}(-1)\, Q_1(1/2).$$

Ejercicio propuesto

8. Realizar operaciones elementales para calcular la inversa de la matriz

$$A = \begin{pmatrix} 1 & -1 & 1 \\ 1 & 1 & 1 \\ 2 & 1 & 0 \end{pmatrix}.$$

1.2. Aplicaciones del cálculo matricial

1.2.1. Modelización matricial de problemas

Una compañía química fabrica dos productos activos. Para obtener un gramo de producto B, la compañía cobra 45 céntimos de materiales, 25 céntimos de mano de obra y 15 céntimos de gastos generales. Para obtener un gramo de producto C, la compañía cobra 40 céntimos de materiales, 30 céntimos de mano de obra y 15 céntimos de gastos generales.

- Suponiendo que la compañía recibe los siguientes pedidos:

  1. 300 gramos de B y 150 gramos de C.
  2. 50 gramos de B y 25 gramos de C.

  ¿Cuáles son los costos desglosados de cada pedido?

```
(%i88) Costes: matrix([45,40],[25,30],[15,15]);
       Pedido1: matrix([300],[150]);
       Pedido2: matrix([50],[25]);
```

$$(\%o88) \begin{pmatrix} 45 & 40 \\ 25 & 30 \\ 15 & 15 \end{pmatrix}$$

$$(\%o89) \begin{pmatrix} 300 \\ 150 \end{pmatrix}$$

$$(\%o90) \begin{pmatrix} 50 \\ 25 \end{pmatrix}$$

```
(%i91) Factura1: Costes.Pedido1;
```

$$(\%o91) \begin{pmatrix} 19500 \\ 12000 \\ 6750 \end{pmatrix}$$

```
(%i92) Factura2: Costes.Pedido2;
```

$$(\%o92) \begin{pmatrix} 3250 \\ 2000 \\ 1125 \end{pmatrix}$$

- Si los dos pedidos anteriores han sido realizados por la misma empresa farmaceutica y se van a facturar de forma conjunta, ¿cuáles serán los costos desglosados que aparecerán en dicha factura?

```
(%i93) Pedidos:addcol(Pedido1,Pedido2);
```

$$(\%o93)\quad \begin{pmatrix} 300 & 50 \\ 150 & 25 \end{pmatrix}$$

```
(%i94) Factura_Conjunta:Costes.Pedidos;
```

$$(\%o94)\quad \begin{pmatrix} 19500 & 3250 \\ 12000 & 2000 \\ 6750 & 1125 \end{pmatrix}$$

- Si la empresa farmaceutica anterior hace todos los meses los dos mismos pedidos que aparecen en el primer apartado, ¿cuáles serán los costos anuales desglosados para dicha empresa?

```
(%i95) Factura_Anual:12*Factura_Conjunta;
```

$$(\%o95)\quad \begin{pmatrix} 234000 & 39000 \\ 144000 & 24000 \\ 81000 & 13500 \end{pmatrix}$$

Ejercicio propuesto

9. Dos tiendas (T_1, T_2) reciben diariamente memorias USB (P_1) y discos duros (P_2) producidos por dos fabricantes distintos (F_1, F_2). Las ventas de estos productos y el beneficio en euros por cada producto vendido, en cada tienda, se especifica en las siguientes tablas:

Ventas	P_1	P_2
F_1	40	50
F_2	70	80

Beneficios	T_1	T_2
P_1	20	25
P_2	90	80

   a) Representar matricialmente los datos del problema.
   b) Utilizar el producto matricial para hallar una matriz $A = (a_{ij})$ donde a_{ij} represente el beneficio que se obtiene por la venta de los productos del fabricante F_i en la tienda T_j.

1.2.2. Codificación de mensajes

En esta sección se presenta una aplicación del Álgebra Matricial a la Criptografía. Concretamente se muestra un procedimiento biunívoco que utiliza matrices regulares, de manera que a cada carácter del mensaje a codificar le asocia otro carácter encriptado. A continuación describimos los pasos a realizar para cifrar un mensaje.

PASO 1. Asociar a cada letra un número y el 0 para especificar un espacio en blanco.

	a	b	c	d	e	f	g	h	i	j	k	l	m
↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓
0	1	2	3	4	5	6	7	8	9	10	11	12	13

n	ñ	o	p	q	r	s	t	u	v	w	x	y	z
↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓
14	15	16	17	18	19	20	21	22	23	24	25	26	27

```
(%i96) a:1$  b:2$  c:3$  d:4$  e:5$  f:6$  g:7$  h:8$  i:9$  j:10$
       k:11$ l:12$ m:13$ n:14$ ñ:15$ o:16$ p:17$ q:18$ r:19$ s:20$
       t:21$ u:22$ v:23$ w:24$ x:25$ y:26$ z:27$
```

Ejemplo de mensaje a codificar

m	e	n	s	a	j	e		o	r	i	g	i	n	a	l
↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓	↓
13	5	14	20	1	10	5	0	16	19	9	7	9	14	1	12

```
(%i123)original: matrix([m,e,n,s,a,j,e,0,o,r,i,g,i,n,a,l])$
```

Paso 2. Definir la matriz $A \in \mathcal{M}_n(\mathbb{R})$ regular que se utiliza en el cifrado del mensaje. Por ejemplo, consideramos

```
(%i124)A: matrix([1,2,1,0],[-1,1,1,0],[0,1,1,2],[1,0,1,1]);
```

$$(\%o124)\begin{pmatrix} 1 & 2 & 1 & 0 \\ -1 & 1 & 1 & 0 \\ 0 & 1 & 1 & 2 \\ 1 & 0 & 1 & 1 \end{pmatrix}$$

Paso 3. A partir de la sucesión de números resultante de asociar a cada letra del mensaje inicial un número, formar vectores de n componentes X_1, $X_2, \ldots$.

X1				X2				X3				X4			
13	5	14	20	1	10	5	0	16	19	9	7	9	14	1	2

y encriptarlos multiplicando por la matriz A, esto es,

$$X_1 = (13 \;\; 5 \;\; 14 \;\; 20), \qquad Y1 = X_1 A = (28 \;\; 45 \;\; 52 \;\; 48) \ldots$$

```
(%i125)x1:[orig[1,1],orig[1,2],orig[1,3],orig[1,4]]$
       y1:x1.A;
```

$(\%o126)\begin{pmatrix} 28 & 45 & 52 & 48 \end{pmatrix}$

```
(%i127)x2:[orig[1,5],orig[1,6],orig[1,7],orig[1,8]]$
       y2:x2.A;
```

$(\%o128)\begin{pmatrix} -9 & 17 & 16 & 10 \end{pmatrix}$

```
(%i129)x3:[orig[1,9],orig[1,10],orig[1,11],orig[1,12]]$
       y3:x3.A;
```

$(\%o130)\begin{pmatrix} 4 & 60 & 51 & 25 \end{pmatrix}$

```
(%i131)x4:[orig[1,13],orig[1,14],orig[1,15],orig[1,16]]$
       y4:x4.A;
```

$(\%o132)\begin{pmatrix} 7 & 33 & 36 & 14 \end{pmatrix}$

Mensaje codificado

Y1				Y2				Y3				Y4			
28	45	52	48	-9	17	16	10	4	60	51	25	7	33	36	14

Ejercicio propuesto

10. Acabamos de recibir el siguiente mensaje cifrado:

Y1				Y2				Y3				Y4				Y5			
-3	35	29	28	11	13	21	12	20	19	37	53	7	26	34	51	2	7	4	0

Sabiendo que el emisor ha utilizado la misma matriz de codificación que la de la Sección 1.2.2, descifrar el mensaje recibido.
